

Example: ANSYS and 3D element (solid45)

In this example, we revisit problem #3 of homework 5a. This problem will now be solved using a 8-node 3D element (solid45) rather than the beam (beam3) element. Input commands for this problem are show below. Students are encouraged to consult the ANSYS online help on solid45 element for its features and limitations.

```
/prep7
et,1,45                !solid45: 8-node 3D element
mp,ex,1,66e9          !modulus of elasticity
mp,prxy,1,0.3         !poisson ratio

k,1,0,0,0             !keypoints
k,2,0.025/2,0,0
k,3,0.075/2,0,0
k,4,0.075/2,0.025,0
k,5,0.025/2,0.025,0
k,6,0.025/2,0.1,0
k,7,0,0.1,0
k,8,0,0.025,0

l,1,2,1               !create lines from 2 keypoints.
l,2,3,2               !third number represent the number of divisions along the line.
l,3,4,2
l,4,5,2
l,5,6,8
l,6,7,1
l,7,8,8
l,8,1,2
l,5,8,1
l,2,5,2

a,1,2,5,8             !create areas using 4 keypoints. Keypoints must be in either
a,2,3,4,5             !clockwise or counter clockwise order.
a,5,6,7,8

esize,,30            !define number of division for the depth.
!esize command must be issued prior to vext command
vext,1,3,1,,3        !extruding the areas parallel to global z-axis to create volumes
!vext,first area,last area,increment,x,y,z
vmesh,all             !mesh all volumes

nset,s,loc,x,0,0      !select a new set of nodes from x = 0 to x = 0
dsym,symm,x           !apply symmetry
nset,s,loc,z,0,0      !select a new set of nodes from z = 0 to z = 0
nset,r,loc,y,0,0      !select nodes from the previous set from y = 0 to y = 0
d,all,all,0           !constain ux, uy, uz
nset,s,loc,z,3,3      !select nodes on the other end of the beam
nset,r,loc,y,0,0
d,all,ux,0            !constrain ux
d,all,uy,0            !constrain uy
nset,s,loc,z,1.2,1.2  !select node on the top of the beam
nset,r,loc,x,0,0
nset,r,loc,y,0.1,0.1
f,all,fy,-5400/2      !apply load
nset,all              !reselect all nodes
fini
```

```

/solu
solve
fini

```

```

/post1
!path,1,7
!pdef,sigbot,s,z
!path,381,397
!pdef,sigtop,s,z
!plpath,sigtop
!plpath,sigbot
fini

```

```

!create path between node 1 and 7
!store sz under sigbot for this path
!create path between node 381 and 397
!store sz under sigtop for this path
!plot sigtop
!plot sigbot

```

